

*Achievements in answer set programming**

VLADIMIR LIFSCHITZ

University of Texas, Austin, TX, USA
(e-mail: vl@cs.utexas.edu)

submitted 18 July 2017; revised 20 July 2017; accepted 27 July 2017

Abstract

This paper describes an approach to the methodology of answer set programming that can facilitate the design of encodings that are easy to understand and provably correct. Under this approach, after appending a rule or a small group of rules to the emerging program, we include a comment that states what has been “achieved” so far. This strategy allows us to set out our understanding of the design of the program by describing the roles of small parts of the program in a mathematically precise way.

KEYWORDS: answer set programming, teaching logic programming, programming methodology

1 Introduction

This paper describes an approach to the methodology of answer set programming (Marek and Truszczyński 1999; Niemelä 1999) that can facilitate the design of encodings that are easy to understand and provably correct. Under this approach, after appending a rule or a small group of rules to the emerging program, the programmer would include a comment that states what has been “achieved” so far, in a certain precise sense.

Consider, for instance, the following solution to the 8 queens problem, adapted from Gebser *et al.* (2012, Section 3.2)¹.

```
1 % Program 8Queens
2
3 row(1..8).
4 col(1..8).
5 8 { queen(I,J) : col(I), row(J) } 8.
6 :- queen(I,J), queen(I,JJ), J!=JJ.
7 :- queen(I,J), queen(II,J), I!=II.
8 :- queen(I,J), queen(II,JJ), (I,J)!=(II,JJ), |I-II|=|J-JJ|.
```

* This research was partially supported by the National Science Foundation under Grant IIS-1422455

¹ A concise introduction to ASP can be found in Chapter 1 of that book. Examples of programs in this paper are written in the input language of the grounder GRINGO, Version 5.

The first rule of *8Queens* (Line 3), viewed as a one-rule program, has a unique stable model S , which satisfies the following condition:

$$\text{A ground atom of the form } \textit{row}(i) \text{ belongs to } S \text{ iff } i \in \{1, \dots, 8\}. \quad (1.1)$$

Condition (1.1) holds also if S is the stable model of the first two rules of this program. And it holds if S is any stable model of the first three rules, and so on, for all six “prefixes” (initial segments) of the program. This is what we mean by achievement: Once the programmer declares that a property “has been achieved,” he is committed to maintaining this property of stable models until the program is completed.

After writing the second rule (Line 4), the programmer can claim that something else has been achieved:

$$\text{A ground atom of the form } \textit{col}(j) \text{ belongs to } S \text{ iff } j \in \{1, \dots, 8\}. \quad (1.2)$$

This condition holds if S is a stable model of any prefix of the program that includes the first two rules.

Additional properties achieved by adding the third rule can be expressed as follows:

$$\begin{aligned} \text{Set } S \text{ contains exactly 8 ground atoms of the form } \textit{queen}(i, j). \\ \text{For each of these atoms, } i, j \in \{1, \dots, 8\}. \end{aligned} \quad (1.3)$$

If a program is written in this manner, then every achievement documented in the process of writing it describes a property shared by all stable models of the entire program. In some cases, this list of achievements can serve as the skeleton of a proof of its correctness, in the spirit of Edsger Dijkstra’s advice:

... one should not first make the program and then prove its correctness, because then the requirement of providing the proof would only increase the poor programmer’s burden. On the contrary: the programmer should let correctness proof and program grow hand in hand (Dijkstra 1972).

Recording important achievements in the process of writing an ASP program may be similar to recording important loop invariants in procedural programming: It does not ensure the correctness of the program but helps the programmer move toward the goal of proving correctness.

A preliminary report on this project was presented at the 2016 Workshop on Answer Set Programming and Other Computing Paradigms.

2 Programs, prefixes, and achievements

In this paper, by an (ordered) program, we understand a list of rules R_1, \dots, R_n ($n \geq 1$) in the input language of an answer set solver, such as CLINGO (Gebser *et al.* 2015) or DLV (Eiter *et al.* 1998). The order of rules is supposed to reflect the order in which the programmer writes them in the process of creating the program. It does not affect the semantics of the program, but it is essential for understanding the process of programming.

We restrict attention to programs without classical negation. (This limitation is discussed in the conclusion.) Stable models of a program without classical negation are sets of ground atoms that contain no arithmetic operations, intervals, or pools

(Gebser *et al.* 2015, Sections 3.1.7, 3.1.9, 3.1.10). Such ground atoms will be called *precomputed*². An *interpretation* is a set of precomputed atoms.

The k th *prefix* of a program R_1, \dots, R_n , where $1 \leq k \leq n$, is the program R_1, \dots, R_k . We will express that a program Γ is a prefix of a program Π by writing $\Gamma \leq \Pi$. The relation \leq is a total order on the set of prefixes of a program.

An *achievement* of a prefix Γ of Π is a property of sets of interpretations that holds for all stable models of all programs Δ such that $\Gamma \leq \Delta \leq \Pi$. For example, (1.1) is an achievement of the first prefix of program *8Queens*; (1.2) is an achievement of its second prefix; and (1.3) is an achievement of its third prefix. Conditions (1.1) and (1.2), and the conjunction of conditions (1.1)–(1.3), are achievements of the third prefix as well. Any condition that holds for all sets of interpretations is trivially an achievement of any prefix of any program.

The following three conditions are achievements of the last three prefixes of *8Queens*:

Each column of the 8×8 chessboard includes at most one square (i, j) such that the atom *queen*(i, j) belongs to S . (2.1)

Each row of the 8×8 chessboard includes at most one square (i, j) such that the atom *queen*(i, j) belongs to S . (2.2)

Each diagonal of the 8×8 chessboard includes at most one square (i, j) such that the atom *queen*(i, j) belongs to S . (2.3)

Thus every stable model S of *8Queens* satisfies all conditions (1.1)–(2.3).

3 Programs with input

In some programs, constants are used as placeholders for values provided by the user (Gebser *et al.* 2015, Section 3.1.15). For example, the constant n is used as a placeholder for an arbitrary positive integer in the following more general version of *8Queens*:

```

1  % Program NQueens
2
3  row(1..n).
4  col(1..n).
5  n { queen(I,J) : col(I), row(J) } n.
6  :- queen(I,J), queen(I,JJ), J!=JJ.
7  :- queen(I,J), queen(II,J), I!=II.
8  :- queen(I,J), queen(II,JJ), (I,J)!= (II,JJ), |I-II|=|J-JJ|.

```

The value of a placeholder is one kind of input that an answer set solver may expect in addition to the rules of the program. A definition of an “extensional predicate” occurring in the bodies of rules is another kind. Consider, for example,

² This terminology follows Gebser *et al.* (2015, Section 2.1), where “precomputed terms” are defined. Calimeri *et al.* (2012, Section 2.1) talk about elements of the “Herbrand universe” of a program in the same sense.

the following encoding of Hamiltonian cycles, adapted from Gebser *et al.* (2012, Section 3.3):

```

1  % Program Hamiltonian
2
3  1 {in(X,Y) : edge(X,Y) } 1 :- vertex(X).
4  1 {in(X,Y) : edge(X,Y) } 1 :- vertex(Y).
5  reached(X) :- in(v0,X).
6  reached(Y) :- reached(X), in(X,Y).
7  :- not reached(X), vertex(X).

```

It needs to be supplemented by definitions of

- the predicate *vertex*/1, representing the set of vertices of a finite digraph G ,
- the predicate *edge*/2, representing the set of edges of G ,
- the placeholder v_0 , which is a vertex of G .

If \mathbf{i} is a valid input for a program Π , then we can talk about models of Π that are stable for input \mathbf{i} , or “ \mathbf{i} -stable³.”

For a program with input, an achievement is defined as a relation between valid inputs and sets of interpretations. Such a relation R will be called an achievement of a prefix Γ of Π if $R(\mathbf{i}, S)$ holds for every valid input \mathbf{i} and every stable model S of any program Δ such that $\Gamma \leq \Delta \leq \Pi$.

For example, the sentence

$$\text{A ground atom of the form } row(i) \text{ belongs to } S \text{ iff } i \in \{1, \dots, n\} \quad (3.1)$$

expresses a relation between n and S that is an achievement of the first prefix of *NQueens*. It is obtained from condition (1.1) by replacing 8 with n , and achievements of the other prefixes of that program can be obtained in a similar way from conditions (1.2)–(2.3).

The following two conditions are achievements of the first two prefixes of *Hamiltonian*:

$$\begin{aligned} &\text{Every pair } (x, y) \text{ such that the atom } in(x, y) \text{ belongs to } S \text{ is an edge} \\ &\text{of } G; \text{ for every vertex } x \text{ of } G \text{ there is a unique } y \text{ such that the atom} \\ &in(x, y) \text{ belongs to } S. \end{aligned} \quad (3.2)$$

$$\begin{aligned} &\text{For every vertex } y \text{ of } G \text{ there is a unique } x \text{ such that the atom } in(x, y) \\ &\text{belongs to } S. \end{aligned} \quad (3.3)$$

Nothing interesting has been achieved by adding the third rule, but the following condition is an achievement of the fourth prefix of the program:

$$\begin{aligned} &\text{The set of symbols } x \text{ such that the atom } reached(x) \text{ belongs to } S \\ &\text{consists of the vertices } x \text{ for which there exists a walk } v_0, \dots, v_n \text{ such} \\ &\text{that } n \geq 1, v_0 = v_0, v_n = x, \text{ and every atom of the form } in(v_i, v_{i+1}) \\ &\text{belongs to } S. \end{aligned} \quad (3.4)$$

³ Programs with input are similar to lp-functions in the sense of Gelfond and Przymusinska (1996, Section 3). The description of an lp-function specifies not only its input, but also its output; on the other hand, the input of an lp-function includes predicates only, not placeholders.

Finally, here is an achievement of the entire program:

For every vertex x of G , the atom $reached(x)$ belongs to S . (3.5)

4 Records of achievement

A *record of achievement* for a program Π is a function that maps some (possibly all) prefixes of Π , including Π itself, to their achievements. For instance, the function that maps the prefixes of *8Queens* to conditions (1.1)–(2.3) is a record of achievement, as well as the function that maps the first, second, fourth and fifth prefixes of *Hamiltonian* to conditions (3.2)–(3.5). Every program has a trivial record of achievement that maps all its prefixes to the identically true condition.

A record of achievement a can be represented by including in the program, after each prefix Γ in the domain of a , a comment that describes the condition $a(\Gamma)$. Here is the program *NQueens* with a record of achievement encoded in this way by comments, and with an additional comment at the beginning that shows which inputs for the program are considered valid:

```

1  % Program NQueens, with a record of achievement
2
3  % input: size n of the board
4
5  % A square on the board is represented as a pair,
6  % column number and row number, both from the set
7  % {1,...,n}.
8
9  row(1..n).
10 % achieved: row/1 = {1,...,n}.
11
12 col(1..n).
13 % achieved: col/1 = {1,...,n}.
14
15 n { queen(I,J) : col(I), row(J) } n.
16 % achieved: Set queen/2 consists of n squares.
17
18 :- queen(I,J), queen(I,JJ), J!=JJ.
19 % achieved: Each column includes at most one square from
20 %           queen/2.
21
22 :- queen(I,J), queen(II,J), I!=II.
23 % achieved: Each row includes at most one square from
24 %           queen/2.
25
26 :- queen(I,J), queen(II,JJ), (I,J)!= (II,JJ), |I-II|=|J-JJ|.
27 % achieved: Each diagonal includes at most one square from
28 %           queen/2.

```

The comment in Line 10 is a concise reformulation of condition (3.1). It uses $row/1$ as shorthand for “the set of precomputed terms i such that the atom $row(i)$ belongs to S .” In the other comments, $col/1$ and $queen/2$ are understood in a similar way.

Program *Hamiltonian* with a record of achievement below uses another useful convention: In Lines 11 and 12, we understand X and Y as metavariables for precomputed terms. The comment in those lines is a reformulation of condition (3.3).

5 Completeness

If a is a record of achievement for a program Π , and Δ is a prefix of Π , then by $a^*(\Delta)$ we denote the conjunction of conditions $a(\Gamma)$ for all prefixes Γ of Δ that belong to the domain of a . It is clear that condition $a^*(\Delta)$ is an achievement of Δ . In particular, all stable models of Π satisfy condition $a^*(\Pi)$. The converse is, generally, not true—it is possible that an interpretation satisfying $a^*(\Pi)$ is not a stable model of Π .

```

1  % Program Hamiltonian, with a record of achievement
2
3  % input: the set vertex/1 of vertices of a finite digraph G;
4  %         the set edge/2 of edges of G; a vertex v0 of G.
5
6  1 {in(X,Y) : edge(X,Y) } 1 :- vertex(X).
7  % achieved: Set in/2 is a subset of edge/2; for every vertex
8  %         X of G there is a unique Y such that in(X,Y).
9
10 1 {in(X,Y) : edge(X,Y) } 1 :- vertex(Y).
11 % achieved: For every vertex Y of G there is a unique X
12 %         such that in(X,Y).
13
14 reached(X) :- in(v0,X).
15 reached(Y) :- reached(X), in(X,Y).
16 % achieved: Set reached/1 consists of the vertices that are
17 %         reachable from v0 by a path of non-zero length
18 %         in the subgraph of G with the set of edges in/2.
19
20 :- not reached(X), vertex(X).
21 % achieved: reached/1 = vertex/1.

```

The following notation and terminology will be used to discuss this issue. For any prefix Γ of a program Π , by $\text{Preds}(\Gamma)$, we denote the set consisting of the predicates occurring in Γ and the input predicates of Π . For instance, if Γ is the first prefix of *Hamiltonian* (rule in Line 6), then $\text{Preds}(\Pi)$ consists of the predicates

$$\text{vertex}/1, \text{edge}/2, \text{in}/2. \quad (5.1)$$

The expression $\text{Preds}(I)$, where I is an interpretation, stands for the set of predicates occurring in I . About a record of achievement a for a program Π , we will say that it is *complete* if, for every Γ in the domain of a , all interpretations I satisfying $a^*(\Gamma)$ for which $\text{Preds}(I) \subseteq \text{Preds}(\Gamma)$ are stable models of Γ .

The records of achievement for *NQueens* and *Hamiltonian* shown above are complete. The completeness of the latter, for example, entails that

- (a) any interpretation S such that atoms in S contain no predicates other than (5.1) is a stable model of the first rule of *Hamiltonian* iff it satisfies condition (3.2);
- (b) such an interpretation is a stable model of the first two rules of *Hamiltonian* iff it satisfies conditions (3.2) and (3.3);

(c) any interpretation S such that atoms in S contain no predicates other than

vertex/1, edge/2, in/2, reached/1

is a stable model of the first four rules of *Hamiltonian* iff it satisfies conditions (3.2)–(3.4);

(d) such an interpretation is a stable model of the entire program iff it satisfies conditions (3.2)–(3.5).

On the other hand, if we drop the word “unique” from the comment in Lines 7, 8 of that record of achievement then it will become incomplete.

6 Achievement-based answer set programming

Records of achievement in the two examples above are not only complete but also detailed, in the sense that they include achievements for almost all prefixes of the programs. The only rule in these programs that is not followed by an achievement comment is the first rule in the recursive definition of *reached*. The role of that rule cannot be properly explained unless we treat it as part of the definition.

Developing an ASP program along with a complete and detailed record of achievement can be called “achievement-based” answer set programming. This strategy allows us to set out our understanding of the design of the program by describing the roles of individual rules, or small groups of rules, in a mathematically precise way.

One of the advantages of this approach is that comments explaining what is achieved by a group of rules at the beginning of a program help us start testing and debugging it at an early stage, when only a part of the program has been written. To this end, we can run an answer set solver to find some (or all) stable models of the prefix that has been already written and check that they satisfy the conditions in the available “achieved” comments. A mismatch would indicate that there is a bug in the rules of the program written so far, or perhaps that the programmer’s intentions have not been properly documented.

A complete record of achievement is particularly valuable when it is closely related to the program’s specification, because from the completeness of such a record we may be able to conclude that the program is correct. For instance, from property (d) of *Hamiltonian* we can conclude that a set of edges of G is a Hamiltonian cycle iff it has the form $\{(x, y) : in(x, y) \in S\}$ for some stable model S of that program.

To further illustrate the idea of achievement-based ASP, we present below three “real-life” ASP programs accompanied by complete, detailed records of achievement. The first of them, program *SCA* (Brain *et al.* 2012, Figure 1), generates sequence covering arrays⁴ of strength 3. Our version of *SCA* is slightly different from the original program: The constraint in Line 19 here replaces the pair of rules

$$\begin{aligned} \text{hb}(N, X, Z) & :- \text{hb}(N, X, Y), \text{hb}(N, Y, Z). \\ & :- \text{hb}(N, X, X). \end{aligned}$$

⁴ A sequence covering array of strength t is an array of permutations of symbols such that every ordering of any t symbols appears as a subsequence of at least one row.

The reason why we chose to make this change is that the first of the two rules above may temporarily destroy the irreflexivity of the relation of hb_N that was true at the previous step, that property is restored by the second rule. That is not in the spirit of the achievement-based approach, which emphasizes the gradual accumulation of properties that we would like to see in the complete program.

The other two examples are program *Borda*, adapted from Charwat and Pfandler (2015, Encoding 1), which encodes the Borda rule for determining the winner in an election with several candidates⁵, and program *OBT*, adapted from Brooks *et al.* (2007, Section 1), which encodes ordered binary trees⁶.

```

1  % Program SCA
2
3  % input: the number s of symbols 1,...,s; the number n of
4  %       rows 1,...,n.
5
6  sym(1..s).
7  % achieved: sym/1 = {1,...,s}.
8
9  row(1..n).
10 % achieved: row/1 = {1,...,n}.
11
12 1 {hb(N,X,Y); hb(N,Y,X)} 1 :- row(N), sym(X), sym(Y), X!=Y.
13 % For every row N, let hb_N be the binary relation on sym/1
14 % defined by the condition: X hb_N Y iff hb(N,X,Y).
15 % achieved: each relation hb_N is irreflexive; each pair of
16 %       distinct symbols satisfies either X hb_N Y
17 %       or Y hb_N X.
18
19 :- hb(N,X,Y), hb(N,Y,Z), not hb(N,X,Z).
20 % achieved: each relation hb_N is transitive.
21
22 covered(X,Y,Z) :- hb(N,X,Y), hb(N,Y,Z).
23 % For every row N and every symbol X, by M_{N,X} we denote
24 % the symbol that is the X-th smallest w.r.t. hb_N.
25 % achieved: for any symbols X, Y, Z, covered(X,Y,Z) iff,
26 %       for some row N, (X,Y,Z) is a subsequence of
27 %       (M_{N,1},...,M_{N,s}).
28
29 :- not covered(X,Y,Z), sym(X), sym(Y), sym(Z),
30     X!=Y, Y!=Z, X!=Z.
31 % achieved: covered(X,Y,Z) for any pairwise distinct symbols
32 %       X, Y, Z.

```

⁵ Each voter ranks the list of candidates in order of preference. The candidate ranked last gets zero points; next to last gets one point, and so on. The candidate with the most points is the winner.

⁶ An ordered binary tree is a rooted binary tree with the leaves $0, \dots, k$ and internal vertices $k+1, \dots, 2k$ such that (i) every internal vertex is greater than its children, and (ii) for any two internal vertices x and x_1 , $x > x_1$ iff the maximum of the children of x is greater than the maximum of the children of x_1 .


```

1  % Program Borda
2
3  % input: the number m of candidates 1,...,m in an election
4  %       E; the set p/3 of triples (P,Pos,C) such that, for
5  %       a fixed ordering pr_1,...,pr_l of the distinct
6  %       preference relations in the profile of E, candidate
7  %       C is at position Pos in relation pr_P; the set
8  %       votecount/2 of pairs (P,VC) such that relation pr_P
9  %       occurs VC times in the profile of E.
10
11 candidate(1..m).
12 % achieved: candidate/1 = {1,...,m}.
13
14 posScore(P,C,X*VC) :- p(P,Pos,C), X=m-Pos, votecount(P,VC).
15 % achieved: posScore(P,C,S) iff the voters who chose
16 %           relation pr_P in election E contributed S points
17 %           to candidate C under the Borda rule.
18
19 score(C,N) :- candidate(C), N=#sum{S:posScore(_,C,S)}.
20 % achieved: score(C,N) iff candidate C earned N points in
21 %           election E under the Borda rule.
22
23 winner(C) :- score(C,M), M=#max{S:score(_,S)}.
24 % achieved: winner(C) iff the number of points earned by
25 %           candidate C in election E is maximal among all
26 %           candidates.

```

```

1  % Program OBT
2
3  % input: positive integer k.
4
5  leaf(0..k).
6  % achieved: leaf/1 = {0,...,k}.
7
8  vertex(0..2*k).
9  % achieved: vertex/1 = {0,...,2k}.
10
11 internal(X) :- vertex(X), not leaf(X).
12 % achieved: internal/1 = {k+1,...,2k}.
13
14 2 {edge(X,Y) : vertex(Y), X>Y} 2 :- internal(X).
15 % Let G be the digraph with the vertices vertex/1 and the
16 % edges edge/2.
17 % achieved: for every edge (X,Y) of G, X>Y; the out-degree
18 %           of a vertex X in G is 2 if internal(X), and 0
19 %           if leaf(X).
20
21 reachable(X,Y) :- edge(X,Y).
22 reachable(X,Y) :- edge(X,Z), reachable(Z,Y).
23 % achieved: reachable(X,Y) iff Y is reachable from X in G
24 %           by a path of non-zero length.
25
26 :- vertex(X), X!=2*k, not reachable(2*k,X).
27 % achieved: every vertex of G other than 2k is reachable
28 %           from 2k by a path of non-zero length.
29
30 :- reachable(X,X), vertex(X).
31 % achieved: G is acyclic.
32
33 max_child(X,Y) :- edge(X,Y), edge(X,Y1), Y > Y1.
34 % achieved: max_child(X,Y) iff Y is the largest child of X
35 %           in G.
36
37 Y<Y1 :- max_child(X,Y), max_child(X1,Y1), Y>Y1, X<X1.
38 % achieved: for any vertices X, X1 of G such that X<X1, the
39 %           largest child of X is smaller than the largest
40 %           child of X1.

```

7 Achievements in teaching

The achievement-based approach was emphasized in a class on answer set programming taught recently to a group of over 50 undergraduates at the University of Texas at Austin. The idea of an achievement was explained more informally than in this paper, but many examples were given. In most solutions to programming assignments submitted for grading, students attempted to imitate the instructor's use of "input" and "achieved" comments, even though they were not instructed to do

that. The degree of their success depended, of course, on their previous exposure to logic and mathematics. When ASP programs written by students were discussed in class, the instructor emphasized the difference between the correctness of the program on the one hand, and the clarity and correctness of “input” and “achieved” comments on the other.

Comments of these kinds can be used in exercises and test problems. In one case, students were shown an “incomplete listing” of a graph coloring program:

```

1  % Color the vertices of a graph so that no two adjacent
2  % vertices share the same color.
3
4  % input: set vertex/1 of vertices of a graph G;
5  %       set edge/2 of edges of G; set color/1 of colors.
6
7  1 {color(X,C) : color(C)} 1 :- vertex(X).
8  % achieved: for every vertex X there is a unique color C
9  %       such that color(X,C).
10
11 -----
12 % achieved: no two adjacent vertices share the same color.
13
14 #show color/2.

```

The question was, “What rule would you place in Line 11?” On another occasion, students were asked to write a one-rule program for which the following comments would be appropriate:

```

1  % Calculate the number of classes taught today on each of
2  % the seven floors of the computer science building.
3
4  % input: set where/2 of all pairs (C,I) such that class C
5  %       is taught on the I-th floor.
6
7  -----
8  % achieved: howmany(I,N) iff the number of classes taught
9  %       on the I-th floor is N.
10
11 #show howmany/2.

```

8 Conclusion

In achievement-based ASP, we start writing a program by describing its inputs. Then, after every rule or small group of rules, we include a comment describing what has been achieved. Collectively these comments represent a complete, detailed record of achievement.

As we are adding rules to an emerging ASP program, we deal at every step with a single executable piece of code, unlike the non-executable pseudo-code formed in the process of stepwise refinement of a procedural program, and unlike a collection

of executable subroutines formed in the process of bottom-up design. In the process, we think of prefixes of the emerging program as if they were complete programs. We describe their stable models in a way that relates them to the stable models of the final product.

The programs discussed in this paper do not use classical negation (Gelfond and Lifschitz 1990). In the presence of classical negation, answer sets consist of “precomputed classical literals”—precomputed atoms and classical negations of such atoms. Extending the definition of a complete record of achievement to such programs is straightforward. On the other hand, many programs with classical negation contain defaults (Gelfond and Kahl 2014, Chap. 5), such as the closed world assumption and the commonsense law of inertia, and the achievement-based approach may be not so useful in application to programs containing defaults. A default does not “achieve” anything in the technical sense of Section 2.

When ASP is used for representing dynamic domains, a very different methodology can be recommended: First describe the domain in an action description language, and then translate its causal laws into answer set programming (Gelfond and Lifschitz 1993; Lifschitz and Turner 1999), (Gelfond and Kahl 2014, Chap. 8).

According to Gebser *et al.* (2012, Section 3.2),

[t]he basic approach to writing encodings in ASP follows a generate-and-test methodology, also referred to as guess-and-check... A “generating” part is meant to non-deterministically provide solution candidates, while a “testing” part eliminates candidates violating some requirements... Both parts are usually amended by “defining” parts providing auxiliary concepts.

Most programs discussed in this paper are designed in accordance with this basic approach⁷. The advice to keep track of what has been achieved as you are adding rules to your program differs from the “generate-and-test” advice in that it refers to mathematical properties of stable models, and not to programmer’s intentions.

Acknowledgements

Thanks to Michael Gelfond, Amelia Harrison, Yuliya Lierler, Julian Michael, Liangkun Zhao, and the anonymous referees for comments on earlier versions of this paper. Conversations and exchanges of email messages with Mark Denecker, Esra Erdem, Martin Gebser, Roland Kaminski, Johannes Oetsch, Dhananjay Raju, and Mirek Truszczyński helped the author develop a better understanding of the methodology of answer set programming.

⁷ Program *Borda* is an exception—it has no generating part and no testing part. Also, it is not clear whether the designers of *Hamiltonian* intended the second rule for the generating part or for the testing part. (The second rule is syntactically similar to the first, which is definitely a generate rule. On the other hand, adding the second rule does not really generate new solution candidates; it eliminates some of the candidates generated earlier.)

References

- BRAIN, M., ERDEM, E., INOUE, K., OETSCH, J., PÜHRER, J., TOMPITS, H. AND YILMAZ, C. 2012. Event-sequence testing using answer-set programming. *International Journal on Advances in Software* 5, 237–251.
- BROOKS, D. R., ERDEM, E., ERDOĞAN, S. T., MINETT, J. W., AND RINGE, D. 2007. Inferring phylogenetic trees using answer set programming. *Journal of Automated Reasoning* 39, 471–511.
- CALIMERI, F., FABER, W., GEBSER, M., IANNI, G., KAMINSKI, R., KRENNWALLNER, T., LEONE, N., RICCA, F. AND SCHAUB, T. 2012. ASP-Core-2: Input language format [online]. URL: <https://www.mat.unical.it/aspcomp2013/files/ASP-CORE-2.03b.pdf> [Accessed on August 15, 2017].
- CHARWAT, G. AND PFANDLER, A. 2015. Democratix: A declarative approach to winner determination. In *Proc. of the 4th International Conference on Algorithmic Decision Theory (ADT)*.
- DIJKSTRA, E. W. 1972. The humble programmer. *Communications of the ACM* 15, 859–866.
- EITER, T., LEONE, N., MATEIS, C., PFEIFER, G. AND SCARCELLO, F. 1998. The KR system dlV: Progress report, comparisons and benchmarks. In *Proc. of International Conference on Principles of Knowledge Representation and Reasoning (KR)*, A. Cohn, L. Schubert, and S. Shapiro, Eds. 406–417.
- GEBSER, M., HARRISON, A., KAMINSKI, R., LIFSCHITZ, V. AND SCHAUB, T. 2015. Abstract gringo. *Theory and Practice of Logic Programming* 15, 449–463.
- GEBSER, M., KAMINSKI, R., KAUFMANN, B., LINDAUER, M., OSTROWSKI, M., ROMERO, J., SCHAUB, T. AND THIELE, S. 2015. Potassco User Guide, version 2.0 [online]. URL: <https://sourceforge.net/projects/potassco/files/guide/2.0/guide-2.0.pdf/download>.
- GEBSER, M., KAMINSKI, R., KAUFMANN, B. AND SCHAUB, T. 2012. *Answer Set Solving in Practice*. Synthesis Lectures on Artificial Intelligence and Machine Learning. Morgan and Claypool Publishers.
- GELFOND, M. AND KAHL, Y. 2014. *Knowledge Representation, Reasoning, and the Design of Intelligent Agents: The Answer-Set Programming Approach*. Cambridge University Press.
- GELFOND, M. AND LIFSCHITZ, V. 1990. Logic programs with classical negation. In *Proc. of International Conference on Logic Programming (ICLP)*, D. Warren and P. Szeredi, Eds. 579–597.
- GELFOND, M. AND LIFSCHITZ, V. 1993. Representing action and change by logic programs. *Journal of Logic Programming* 17, 301–322.
- GELFOND, M. AND PRZYMUSINSKA, H. 1996. Towards a theory of elaboration tolerance: Logic programming approach. *International Journal of Software Engineering and Knowledge Engineering* 6, 1, 89–112.
- LIFSCHITZ, V. AND TURNER, H. 1999. Representing transition systems by logic programs. In *Proc. of International Conference on Logic Programming and Nonmonotonic Reasoning (LPNMR)*. 92–106.
- MAREK, V. AND TRUSZCZYNSKI, M. 1999. Stable models and an alternative logic programming paradigm. In *The Logic Programming Paradigm: A 25-Year Perspective*. Springer Verlag, 375–398.
- NIEMELÄ, I. 1999. Logic programs with stable model semantics as a constraint programming paradigm. *Annals of Mathematics and Artificial Intelligence* 25, 241–273.